creation_sourcecode: // Sources flattened with hardhat v2.20.1 https://hardhat.org
  
  // SPDX-License-Identifier: MIT AND UNLICENSED
  
  // File @openzeppelin/contracts/access/IAccessControl.sol@v5.0.1
  
  // Original license: SPDX_License_Identifier: MIT
  // OpenZeppelin Contracts (last updated v5.0.0) (access/IAccessControl.sol)
  
  pragma solidity ^0.8.20;
  
  /**
   * @dev External interface of AccessControl declared to support ERC165 detection.
   */
  interface IAccessControl {
      /**
       * @dev The `account` is missing a role.
       */
      error AccessControlUnauthorizedAccount(address account, bytes32 neededRole);
  
      /**
       * @dev The caller of a function is not the expected one.
       *
       * NOTE: Don't confuse with {AccessControlUnauthorizedAccount}.
       */
      error AccessControlBadConfirmation();
  
      /**
       * @dev Emitted when `newAdminRole` is set as ``role``'s admin role, replacing `previousAdminRole`
       *
       * `DEFAULT_ADMIN_ROLE` is the starting admin for all roles, despite
       * {RoleAdminChanged} not being emitted signaling this.
       */
      event RoleAdminChanged(bytes32 indexed role, bytes32 indexed previousAdminRole, bytes32 indexed newAdminRole);
  
      /**
       * @dev Emitted when `account` is granted `role`.
       *
       * `sender` is the account that originated the contract call, an admin role
       * bearer except when using {AccessControl-_setupRole}.
       */
      event RoleGranted(bytes32 indexed role, address indexed account, address indexed sender);
  
      /**
       * @dev Emitted when `account` is revoked `role`.
       *
       * `sender` is the account that originated the contract call:
       *   - if using `revokeRole`, it is the admin role bearer
       *   - if using `renounceRole`, it is the role bearer (i.e. `account`)
       */
      event RoleRevoked(bytes32 indexed role, address indexed account, address indexed sender);
  
      /**
       * @dev Returns `true` if `account` has been granted `role`.
       */
      function hasRole(bytes32 role, address account) external view returns (bool);
  
      /**
       * @dev Returns the admin role that controls `role`. See {grantRole} and
       * {revokeRole}.
       *
       * To change a role's admin, use {AccessControl-_setRoleAdmin}.
       */
      function getRoleAdmin(bytes32 role) external view returns (bytes32);
  
      /**
       * @dev Grants `role` to `account`.
       *
       * If `account` had not been already granted `role`, emits a {RoleGranted}
       * event.
       *
       * Requirements:
       *
       * - the caller must have ``role``'s admin role.
       */
      function grantRole(bytes32 role, address account) external;
  
      /**
       * @dev Revokes `role` from `account`.
       *
       * If `account` had been granted `role`, emits a {RoleRevoked} event.
       *
       * Requirements:
       *
       * - the caller must have ``role``'s admin role.
       */
      function revokeRole(bytes32 role, address account) external;
  
      /**
       * @dev Revokes `role` from the calling account.
       *
       * Roles are often managed via {grantRole} and {revokeRole}: this function's
       * purpose is to provide a mechanism for accounts to lose their privileges
       * if they are compromised (such as when a trusted device is misplaced).
       *
       * If the calling account had been granted `role`, emits a {RoleRevoked}
       * event.
       *
       * Requirements:
       *
       * - the caller must be `callerConfirmation`.
       */
      function renounceRole(bytes32 role, address callerConfirmation) external;
  }
  
  
  // File @openzeppelin/contracts/utils/Context.sol@v5.0.1
  
  // Original license: SPDX_License_Identifier: MIT
  // OpenZeppelin Contracts (last updated v5.0.1) (utils/Context.sol)
  
  pragma solidity ^0.8.20;
  
  /**
   * @dev Provides information about the current execution context, including the
   * sender of the transaction and its data. While these are generally available
   * via msg.sender and msg.data, they should not be accessed in such a direct
   * manner, since when dealing with meta-transactions the account sending and
   * paying for execution may not be the actual sender (as far as an application
   * is concerned).
   *
   * This contract is only required for intermediate, library-like contracts.
   */
  abstract contract Context {
      function _msgSender() internal view virtual returns (address) {
          return msg.sender;
      }
  
      function _msgData() internal view virtual returns (bytes calldata) {
          return msg.data;
      }
  
      function _contextSuffixLength() internal view virtual returns (uint256) {
          return 0;
      }
  }
  
  
  // File @openzeppelin/contracts/utils/introspection/IERC165.sol@v5.0.1
  
  // Original license: SPDX_License_Identifier: MIT
  // OpenZeppelin Contracts (last updated v5.0.0) (utils/introspection/IERC165.sol)
  
  pragma solidity ^0.8.20;
  
  /**
   * @dev Interface of the ERC165 standard, as defined in the
   * https://eips.ethereum.org/EIPS/eip-165[EIP].
   *
   * Implementers can declare support of contract interfaces, which can then be
   * queried by others ({ERC165Checker}).
   *
   * For an implementation, see {ERC165}.
   */
  interface IERC165 {
      /**
       * @dev Returns true if this contract implements the interface defined by
       * `interfaceId`. See the corresponding
       * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
       * to learn more about how these ids are created.
       *
       * This function call must use less than 30 000 gas.
       */
      function supportsInterface(bytes4 interfaceId) external view returns (bool);
  }
  
  
  // File @openzeppelin/contracts/utils/introspection/ERC165.sol@v5.0.1
  
  // Original license: SPDX_License_Identifier: MIT
  // OpenZeppelin Contracts (last updated v5.0.0) (utils/introspection/ERC165.sol)
  
  pragma solidity ^0.8.20;
  
  /**
   * @dev Implementation of the {IERC165} interface.
   *
   * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
   * for the additional interface id that will be supported. For example:
   *
   * ```solidity
   * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
   *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
   * }
   * ```
   */
  abstract contract ERC165 is IERC165 {
      /**
       * @dev See {IERC165-supportsInterface}.
       */
      function supportsInterface(bytes4 interfaceId) public view virtual returns (bool) {
          return interfaceId == type(IERC165).interfaceId;
      }
  }
  
  
  // File @openzeppelin/contracts/access/AccessControl.sol@v5.0.1
  
  // Original license: SPDX_License_Identifier: MIT
  // OpenZeppelin Contracts (last updated v5.0.0) (access/AccessControl.sol)
  
  pragma solidity ^0.8.20;
  
  
  
  /**
   * @dev Contract module that allows children to implement role-based access
   * control mechanisms. This is a lightweight version that doesn't allow enumerating role
   * members except through off-chain means by accessing the contract event logs. Some
   * applications may benefit from on-chain enumerability, for those cases see
   * {AccessControlEnumerable}.
   *
   * Roles are referred to by their `bytes32` identifier. These should be exposed
   * in the external API and be unique. The best way to achieve this is by
   * using `public constant` hash digests:
   *
   * ```solidity
   * bytes32 public constant MY_ROLE = keccak256("MY_ROLE");
   * ```
   *
   * Roles can be used to represent a set of permissions. To restrict access to a
   * function call, use {hasRole}:
   *
   * ```solidity
   * function foo() public {
   *     require(hasRole(MY_ROLE, msg.sender));
   *     ...
   * }
   * ```
   *
   * Roles can be granted and revoked dynamically via the {grantRole} and
   * {revokeRole} functions. Each role has an associated admin role, and only
   * accounts that have a role's admin role can call {grantRole} and {revokeRole}.
   *
   * By default, the admin role for all roles is `DEFAULT_ADMIN_ROLE`, which means
   * that only accounts with this role will be able to grant or revoke other
   * roles. More complex role relationships can be created by using
   * {_setRoleAdmin}.
   *
   * WARNING: The `DEFAULT_ADMIN_ROLE` is also its own admin: it has permission to
   * grant and revoke this role. Extra precautions should be taken to secure
   * accounts that have been granted it. We recommend using {AccessControlDefaultAdminRules}
   * to enforce additional security measures for this role.
   */
  abstract contract AccessControl is Context, IAccessControl, ERC165 {
      struct RoleData {
          mapping(address account => bool) hasRole;
          bytes32 adminRole;
      }
  
      mapping(bytes32 role => RoleData) private _roles;
  
      bytes32 public constant DEFAULT_ADMIN_ROLE = 0x00;
  
      /**
       * @dev Modifier that checks that an account has a specific role. Reverts
       * with an {AccessControlUnauthorizedAccount} error including the required role.
       */
      modifier onlyRole(bytes32 role) {
          _checkRole(role);
          _;
      }
  
      /**
       * @dev See {IERC165-supportsInterface}.
       */
      function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
          return interfaceId == type(IAccessControl).interfaceId || super.supportsInterface(interfaceId);
      }
  
      /**
       * @dev Returns `true` if `account` has been granted `role`.
       */
      function hasRole(bytes32 role, address account) public view virtual returns (bool) {
          return _roles[role].hasRole[account];
      }
  
      /**
       * @dev Reverts with an {AccessControlUnauthorizedAccount} error if `_msgSender()`
       * is missing `role`. Overriding this function changes the behavior of the {onlyRole} modifier.
       */
      function _checkRole(bytes32 role) internal view virtual {
          _checkRole(role, _msgSender());
      }
  
      /**
       * @dev Reverts with an {AccessControlUnauthorizedAccount} error if `account`
       * is missing `role`.
       */
      function _checkRole(bytes32 role, address account) internal view virtual {
          if (!hasRole(role, account)) {
              revert AccessControlUnauthorizedAccount(account, role);
          }
      }
  
      /**
       * @dev Returns the admin role that controls `role`. See {grantRole} and
       * {revokeRole}.
       *
       * To change a role's admin, use {_setRoleAdmin}.
       */
      function getRoleAdmin(bytes32 role) public view virtual returns (bytes32) {
          return _roles[role].adminRole;
      }
  
      /**
       * @dev Grants `role` to `account`.
       *
       * If `account` had not been already granted `role`, emits a {RoleGranted}
       * event.
       *
       * Requirements:
       *
       * - the caller must have ``role``'s admin role.
       *
       * May emit a {RoleGranted} event.
       */
      function grantRole(bytes32 role, address account) public virtual onlyRole(getRoleAdmin(role)) {
          _grantRole(role, account);
      }
  
      /**
       * @dev Revokes `role` from `account`.
       *
       * If `account` had been granted `role`, emits a {RoleRevoked} event.
       *
       * Requirements:
       *
       * - the caller must have ``role``'s admin role.
       *
       * May emit a {RoleRevoked} event.
       */
      function revokeRole(bytes32 role, address account) public virtual onlyRole(getRoleAdmin(role)) {
          _revokeRole(role, account);
      }
  
      /**
       * @dev Revokes `role` from the calling account.
       *
       * Roles are often managed via {grantRole} and {revokeRole}: this function's
       * purpose is to provide a mechanism for accounts to lose their privileges
       * if they are compromised (such as when a trusted device is misplaced).
       *
       * If the calling account had been revoked `role`, emits a {RoleRevoked}
       * event.
       *
       * Requirements:
       *
       * - the caller must be `callerConfirmation`.
       *
       * May emit a {RoleRevoked} event.
       */
      function renounceRole(bytes32 role, address callerConfirmation) public virtual {
          if (callerConfirmation != _msgSender()) {
              revert AccessControlBadConfirmation();
          }
  
          _revokeRole(role, callerConfirmation);
      }
  
      /**
       * @dev Sets `adminRole` as ``role``'s admin role.
       *
       * Emits a {RoleAdminChanged} event.
       */
      function _setRoleAdmin(bytes32 role, bytes32 adminRole) internal virtual {
          bytes32 previousAdminRole = getRoleAdmin(role);
          _roles[role].adminRole = adminRole;
          emit RoleAdminChanged(role, previousAdminRole, adminRole);
      }
  
      /**
       * @dev Attempts to grant `role` to `account` and returns a boolean indicating if `role` was granted.
       *
       * Internal function without access restriction.
       *
       * May emit a {RoleGranted} event.
       */
      function _grantRole(bytes32 role, address account) internal virtual returns (bool) {
          if (!hasRole(role, account)) {
              _roles[role].hasRole[account] = true;
              emit RoleGranted(role, account, _msgSender());
              return true;
          } else {
              return false;
          }
      }
  
      /**
       * @dev Attempts to revoke `role` to `account` and returns a boolean indicating if `role` was revoked.
       *
       * Internal function without access restriction.
       *
       * May emit a {RoleRevoked} event.
       */
      function _revokeRole(bytes32 role, address account) internal virtual returns (bool) {
          if (hasRole(role, account)) {
              _roles[role].hasRole[account] = false;
              emit RoleRevoked(role, account, _msgSender());
              return true;
          } else {
              return false;
          }
      }
  }
  
  
  // File @openzeppelin/contracts/token/ERC20/extensions/IERC20Permit.sol@v5.0.1
  
  // Original license: SPDX_License_Identifier: MIT
  // OpenZeppelin Contracts (last updated v5.0.0) (token/ERC20/extensions/IERC20Permit.sol)
  
  pragma solidity ^0.8.20;
  
  /**
   * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
   * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
   *
   * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
   * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
   * need to send a transaction, and thus is not required to hold Ether at all.
   *
   * ==== Security Considerations
   *
   * There are two important considerations concerning the use of `permit`. The first is that a valid permit signature
   * expresses an allowance, and it should not be assumed to convey additional meaning. In particular, it should not be
   * considered as an intention to spend the allowance in any specific way. The second is that because permits have
   * built-in replay protection and can be submitted by anyone, they can be frontrun. A protocol that uses permits should
   * take this into consideration and allow a `permit` call to fail. Combining these two aspects, a pattern that may be
   * generally recommended is:
   *
   * ```solidity
   * function doThingWithPermit(..., uint256 value, uint256 deadline, uint8 v, bytes32 r, bytes32 s) public {
   *     try token.permit(msg.sender, address(this), value, deadline, v, r, s) {} catch {}
   *     doThing(..., value);
   * }
   *
   * function doThing(..., uint256 value) public {
   *     token.safeTransferFrom(msg.sender, address(this), value);
   *     ...
   * }
   * ```
   *
   * Observe that: 1) `msg.sender` is used as the owner, leaving no ambiguity as to the signer intent, and 2) the use of
   * `try/catch` allows the permit to fail and makes the code tolerant to frontrunning. (See also
   * {SafeERC20-safeTransferFrom}).
   *
   * Additionally, note that smart contract wallets (such as Argent or Safe) are not able to produce permit signatures, so
   * contracts should have entry points that don't rely on permit.
   */
  interface IERC20Permit {
      /**
       * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
       * given ``owner``'s signed approval.
       *
       * IMPORTANT: The same issues {IERC20-approve} has related to transaction
       * ordering also apply here.
       *
       * Emits an {Approval} event.
       *
       * Requirements:
       *
       * - `spender` cannot be the zero address.
       * - `deadline` must be a timestamp in the future.
       * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
       * over the EIP712-formatted function arguments.
       * - the signature must use ``owner``'s current nonce (see {nonces}).
       *
       * For more information on the signature format, see the
       * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
       * section].
       *
       * CAUTION: See Security Considerations above.
       */
      function permit(
          address owner,
          address spender,
          uint256 value,
          uint256 deadline,
          uint8 v,
          bytes32 r,
          bytes32 s
      ) external;
  
      /**
       * @dev Returns the current nonce for `owner`. This value must be
       * included whenever a signature is generated for {permit}.
       *
       * Every successful call to {permit} increases ``owner``'s nonce by one. This
       * prevents a signature from being used multiple times.
       */
      function nonces(address owner) external view returns (uint256);
  
      /**
       * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
       */
      // solhint-disable-next-line func-name-mixedcase
      function DOMAIN_SEPARATOR() external view returns (bytes32);
  }
  
  
  // File @openzeppelin/contracts/token/ERC20/IERC20.sol@v5.0.1
  
  // Original license: SPDX_License_Identifier: MIT
  // OpenZeppelin Contracts (last updated v5.0.0) (token/ERC20/IERC20.sol)
  
  pragma solidity ^0.8.20;
  
  /**
   * @dev Interface of the ERC20 standard as defined in the EIP.
   */
  interface IERC20 {
      /**
       * @dev Emitted when `value` tokens are moved from one account (`from`) to
       * another (`to`).
       *
       * Note that `value` may be zero.
       */
      event Transfer(address indexed from, address indexed to, uint256 value);
  
      /**
       * @dev Emitted when the allowance of a `spender` for an `owner` is set by
       * a call to {approve}. `value` is the new allowance.
       */
      event Approval(address indexed owner, address indexed spender, uint256 value);
  
      /**
       * @dev Returns the value of tokens in existence.
       */
      function totalSupply() external view returns (uint256);
  
      /**
       * @dev Returns the value of tokens owned by `account`.
       */
      function balanceOf(address account) external view returns (uint256);
  
      /**
       * @dev Moves a `value` amount of tokens from the caller's account to `to`.
       *
       * Returns a boolean value indicating whether the operation succeeded.
       *
       * Emits a {Transfer} event.
       */
      function transfer(address to, uint256 value) external returns (bool);
  
      /**
       * @dev Returns the remaining number of tokens that `spender` will be
       * allowed to spend on behalf of `owner` through {transferFrom}. This is
       * zero by default.
       *
       * This value changes when {approve} or {transferFrom} are called.
       */
      function allowance(address owner, address spender) external view returns (uint256);
  
      /**
       * @dev Sets a `value` amount of tokens as the allowance of `spender` over the
       * caller's tokens.
       *
       * Returns a boolean value indicating whether the operation succeeded.
       *
       * IMPORTANT: Beware that changing an allowance with this method brings the risk
       * that someone may use both the old and the new allowance by unfortunate
       * transaction ordering. One possible solution to mitigate this race
       * condition is to first reduce the spender's allowance to 0 and set the
       * desired value afterwards:
       * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
       *
       * Emits an {Approval} event.
       */
      function approve(address spender, uint256 value) external returns (bool);
  
      /**
       * @dev Moves a `value` amount of tokens from `from` to `to` using the
       * allowance mechanism. `value` is then deducted from the caller's
       * allowance.
       *
       * Returns a boolean value indicating whether the operation succeeded.
       *
       * Emits a {Transfer} event.
       */
      function transferFrom(address from, address to, uint256 value) external returns (bool);
  }
  
  
  // File @openzeppelin/contracts/utils/Address.sol@v5.0.1
  
  // Original license: SPDX_License_Identifier: MIT
  // OpenZeppelin Contracts (last updated v5.0.0) (utils/Address.sol)
  
  pragma solidity ^0.8.20;
  
  /**
   * @dev Collection of functions related to the address type
   */
  library Address {
      /**
       * @dev The ETH balance of the account is not enough to perform the operation.
       */
      error AddressInsufficientBalance(address account);
  
      /**
       * @dev There's no code at `target` (it is not a contract).
       */
      error AddressEmptyCode(address target);
  
      /**
       * @dev A call to an address target failed. The target may have reverted.
       */
      error FailedInnerCall();
  
      /**
       * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
       * `recipient`, forwarding all available gas and reverting on errors.
       *
       * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
       * of certain opcodes, possibly making contracts go over the 2300 gas limit
       * imposed by `transfer`, making them unable to receive funds via
       * `transfer`. {sendValue} removes this limitation.
       *
       * https://consensys.net/diligence/blog/2019/09/stop-using-soliditys-transfer-now/[Learn more].
       *
       * IMPORTANT: because control is transferred to `recipient`, care must be
       * taken to not create reentrancy vulnerabilities. Consider using
       * {ReentrancyGuard} or the
       * https://solidity.readthedocs.io/en/v0.8.20/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
       */
      function sendValue(address payable recipient, uint256 amount) internal {
          if (address(this).balance < amount) {
              revert AddressInsufficientBalance(address(this));
          }
  
          (bool success, ) = recipient.call{value: amount}("");
          if (!success) {
              revert FailedInnerCall();
          }
      }
  
      /**
       * @dev Performs a Solidity function call using a low level `call`. A
       * plain `call` is an unsafe replacement for a function call: use this
       * function instead.
       *
       * If `target` reverts with a revert reason or custom error, it is bubbled
       * up by this function (like regular Solidity function calls). However, if
       * the call reverted with no returned reason, this function reverts with a
       * {FailedInnerCall} error.
       *
       * Returns the raw returned data. To convert to the expected return value,
       * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
       *
       * Requirements:
       *
       * - `target` must be a contract.
       * - calling `target` with `data` must not revert.
       */
      function functionCall(address target, bytes memory data) internal returns (bytes memory) {
          return functionCallWithValue(target, data, 0);
      }
  
      /**
       * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
       * but also transferring `value` wei to `target`.
       *
       * Requirements:
       *
       * - the calling contract must have an ETH balance of at least `value`.
       * - the called Solidity function must be `payable`.
       */
      function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
          if (address(this).balance < value) {
              revert AddressInsufficientBalance(address(this));
          }
          (bool success, bytes memory returndata) = target.call{value: value}(data);
          return verifyCallResultFromTarget(target, success, returndata);
      }
  
      /**
       * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
       * but performing a static call.
       */
      function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
          (bool success, bytes memory returndata) = target.staticcall(data);
          return verifyCallResultFromTarget(target, success, returndata);
      }
  
      /**
       * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
       * but performing a delegate call.
       */
      function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
          (bool success, bytes memory returndata) = target.delegatecall(data);
          return verifyCallResultFromTarget(target, success, returndata);
      }
  
      /**
       * @dev Tool to verify that a low level call to smart-contract was successful, and reverts if the target
       * was not a contract or bubbling up the revert reason (falling back to {FailedInnerCall}) in case of an
       * unsuccessful call.
       */
      function verifyCallResultFromTarget(
          address target,
          bool success,
          bytes memory returndata
      ) internal view returns (bytes memory) {
          if (!success) {
              _revert(returndata);
          } else {
              // only check if target is a contract if the call was successful and the return data is empty
              // otherwise we already know that it was a contract
              if (returndata.length == 0 && target.code.length == 0) {
                  revert AddressEmptyCode(target);
              }
              return returndata;
          }
      }
  
      /**
       * @dev Tool to verify that a low level call was successful, and reverts if it wasn't, either by bubbling the
       * revert reason or with a default {FailedInnerCall} error.
       */
      function verifyCallResult(bool success, bytes memory returndata) internal pure returns (bytes memory) {
          if (!success) {
              _revert(returndata);
          } else {
              return returndata;
          }
      }
  
      /**
       * @dev Reverts with returndata if present. Otherwise reverts with {FailedInnerCall}.
       */
      function _revert(bytes memory returndata) private pure {
          // Look for revert reason and bubble it up if present
          if (returndata.length > 0) {
              // The easiest way to bubble the revert reason is using memory via assembly
              /// @solidity memory-safe-assembly
              assembly {
                  let returndata_size := mload(returndata)
                  revert(add(32, returndata), returndata_size)
              }
          } else {
              revert FailedInnerCall();
          }
      }
  }
  
  
  // File @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol@v5.0.1
  
  // Original license: SPDX_License_Identifier: MIT
  // OpenZeppelin Contracts (last updated v5.0.0) (token/ERC20/utils/SafeERC20.sol)
  
  pragma solidity ^0.8.20;
  
  
  
  /**
   * @title SafeERC20
   * @dev Wrappers around ERC20 operations that throw on failure (when the token
   * contract returns false). Tokens that return no value (and instead revert or
   * throw on failure) are also supported, non-reverting calls are assumed to be
   * successful.
   * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
   * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
   */
  library SafeERC20 {
      using Address for address;
  
      /**
       * @dev An operation with an ERC20 token failed.
       */
      error SafeERC20FailedOperation(address token);
  
      /**
       * @dev Indicates a failed `decreaseAllowance` request.
       */
      error SafeERC20FailedDecreaseAllowance(address spender, uint256 currentAllowance, uint256 requestedDecrease);
  
      /**
       * @dev Transfer `value` amount of `token` from the calling contract to `to`. If `token` returns no value,
       * non-reverting calls are assumed to be successful.
       */
      function safeTransfer(IERC20 token, address to, uint256 value) internal {
          _callOptionalReturn(token, abi.encodeCall(token.transfer, (to, value)));
      }
  
      /**
       * @dev Transfer `value` amount of `token` from `from` to `to`, spending the approval given by `from` to the
       * calling contract. If `token` returns no value, non-reverting calls are assumed to be successful.
       */
      function safeTransferFrom(IERC20 token, address from, address to, uint256 value) internal {
          _callOptionalReturn(token, abi.encodeCall(token.transferFrom, (from, to, value)));
      }
  
      /**
       * @dev Increase the calling contract's allowance toward `spender` by `value`. If `token` returns no value,
       * non-reverting calls are assumed to be successful.
       */
      function safeIncreaseAllowance(IERC20 token, address spender, uint256 value) internal {
          uint256 oldAllowance = token.allowance(address(this), spender);
          forceApprove(token, spender, oldAllowance + value);
      }
  
      /**
       * @dev Decrease the calling contract's allowance toward `spender` by `requestedDecrease`. If `token` returns no
       * value, non-reverting calls are assumed to be successful.
       */
      function safeDecreaseAllowance(IERC20 token, address spender, uint256 requestedDecrease) internal {
          unchecked {
              uint256 currentAllowance = token.allowance(address(this), spender);
              if (currentAllowance < requestedDecrease) {
                  revert SafeERC20FailedDecreaseAllowance(spender, currentAllowance, requestedDecrease);
              }
              forceApprove(token, spender, currentAllowance - requestedDecrease);
          }
      }
  
      /**
       * @dev Set the calling contract's allowance toward `spender` to `value`. If `token` returns no value,
       * non-reverting calls are assumed to be successful. Meant to be used with tokens that require the approval
       * to be set to zero before setting it to a non-zero value, such as USDT.
       */
      function forceApprove(IERC20 token, address spender, uint256 value) internal {
          bytes memory approvalCall = abi.encodeCall(token.approve, (spender, value));
  
          if (!_callOptionalReturnBool(token, approvalCall)) {
              _callOptionalReturn(token, abi.encodeCall(token.approve, (spender, 0)));
              _callOptionalReturn(token, approvalCall);
          }
      }
  
      /**
       * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
       * on the return value: the return value is optional (but if data is returned, it must not be false).
       * @param token The token targeted by the call.
       * @param data The call data (encoded using abi.encode or one of its variants).
       */
      function _callOptionalReturn(IERC20 token, bytes memory data) private {
          // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
          // we're implementing it ourselves. We use {Address-functionCall} to perform this call, which verifies that
          // the target address contains contract code and also asserts for success in the low-level call.
  
          bytes memory returndata = address(token).functionCall(data);
          if (returndata.length != 0 && !abi.decode(returndata, (bool))) {
              revert SafeERC20FailedOperation(address(token));
          }
      }
  
      /**
       * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
       * on the return value: the return value is optional (but if data is returned, it must not be false).
       * @param token The token targeted by the call.
       * @param data The call data (encoded using abi.encode or one of its variants).
       *
       * This is a variant of {_callOptionalReturn} that silents catches all reverts and returns a bool instead.
       */
      function _callOptionalReturnBool(IERC20 token, bytes memory data) private returns (bool) {
          // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
          // we're implementing it ourselves. We cannot use {Address-functionCall} here since this should return false
          // and not revert is the subcall reverts.
  
          (bool success, bytes memory returndata) = address(token).call(data);
          return success && (returndata.length == 0 || abi.decode(returndata, (bool))) && address(token).code.length > 0;
      }
  }
  
  
  // File @openzeppelin/contracts/utils/Multicall.sol@v5.0.1
  
  // Original license: SPDX_License_Identifier: MIT
  // OpenZeppelin Contracts (last updated v5.0.1) (utils/Multicall.sol)
  
  pragma solidity ^0.8.20;
  
  
  /**
   * @dev Provides a function to batch together multiple calls in a single external call.
   *
   * Consider any assumption about calldata validation performed by the sender may be violated if it's not especially
   * careful about sending transactions invoking {multicall}. For example, a relay address that filters function
   * selectors won't filter calls nested within a {multicall} operation.
   *
   * NOTE: Since 5.0.1 and 4.9.4, this contract identifies non-canonical contexts (i.e. `msg.sender` is not {_msgSender}).
   * If a non-canonical context is identified, the following self `delegatecall` appends the last bytes of `msg.data`
   * to the subcall. This makes it safe to use with {ERC2771Context}. Contexts that don't affect the resolution of
   * {_msgSender} are not propagated to subcalls.
   */
  abstract contract Multicall is Context {
      /**
       * @dev Receives and executes a batch of function calls on this contract.
       * @custom:oz-upgrades-unsafe-allow-reachable delegatecall
       */
      function multicall(bytes[] calldata data) external virtual returns (bytes[] memory results) {
          bytes memory context = msg.sender == _msgSender()
              ? new bytes(0)
              : msg.data[msg.data.length - _contextSuffixLength():];
  
          results = new bytes[](data.length);
          for (uint256 i = 0; i < data.length; i++) {
              results[i] = Address.functionDelegateCall(address(this), bytes.concat(data[i], context));
          }
          return results;
      }
  }
  
  
  // File @openzeppelin/contracts/utils/Pausable.sol@v5.0.1
  
  // Original license: SPDX_License_Identifier: MIT
  // OpenZeppelin Contracts (last updated v5.0.0) (utils/Pausable.sol)
  
  pragma solidity ^0.8.20;
  
  /**
   * @dev Contract module which allows children to implement an emergency stop
   * mechanism that can be triggered by an authorized account.
   *
   * This module is used through inheritance. It will make available the
   * modifiers `whenNotPaused` and `whenPaused`, which can be applied to
   * the functions of your contract. Note that they will not be pausable by
   * simply including this module, only once the modifiers are put in place.
   */
  abstract contract Pausable is Context {
      bool private _paused;
  
      /**
       * @dev Emitted when the pause is triggered by `account`.
       */
      event Paused(address account);
  
      /**
       * @dev Emitted when the pause is lifted by `account`.
       */
      event Unpaused(address account);
  
      /**
       * @dev The operation failed because the contract is paused.
       */
      error EnforcedPause();
  
      /**
       * @dev The operation failed because the contract is not paused.
       */
      error ExpectedPause();
  
      /**
       * @dev Initializes the contract in unpaused state.
       */
      constructor() {
          _paused = false;
      }
  
      /**
       * @dev Modifier to make a function callable only when the contract is not paused.
       *
       * Requirements:
       *
       * - The contract must not be paused.
       */
      modifier whenNotPaused() {
          _requireNotPaused();
          _;
      }
  
      /**
       * @dev Modifier to make a function callable only when the contract is paused.
       *
       * Requirements:
       *
       * - The contract must be paused.
       */
      modifier whenPaused() {
          _requirePaused();
          _;
      }
  
      /**
       * @dev Returns true if the contract is paused, and false otherwise.
       */
      function paused() public view virtual returns (bool) {
          return _paused;
      }
  
      /**
       * @dev Throws if the contract is paused.
       */
      function _requireNotPaused() internal view virtual {
          if (paused()) {
              revert EnforcedPause();
          }
      }
  
      /**
       * @dev Throws if the contract is not paused.
       */
      function _requirePaused() internal view virtual {
          if (!paused()) {
              revert ExpectedPause();
          }
      }
  
      /**
       * @dev Triggers stopped state.
       *
       * Requirements:
       *
       * - The contract must not be paused.
       */
      function _pause() internal virtual whenNotPaused {
          _paused = true;
          emit Paused(_msgSender());
      }
  
      /**
       * @dev Returns to normal state.
       *
       * Requirements:
       *
       * - The contract must be paused.
       */
      function _unpause() internal virtual whenPaused {
          _paused = false;
          emit Unpaused(_msgSender());
      }
  }
  
  
  // File contracts/ExchangeBridge.sol
  
  // Original license: SPDX_License_Identifier: UNLICENSED
  pragma solidity 0.8.24;
  
  
  
  
  //TODO slippage values in lock function 10**18
  contract ExchangeBridge is AccessControl, Pausable, Multicall {
      using SafeERC20 for IERC20;
  
      address public liquidityPool;
  
      bytes32 public constant ADMIN_ROLE = keccak256("ADMIN_ROLE");
      bytes32 public constant PAUSER_ROLE = keccak256("PAUSER_ROLE");
      bytes32 public constant UNPAUSER_ROLE = keccak256("UNPAUSER_ROLE");
      bytes32 public constant UNLOCKER_ROLE = keccak256("UNLOCKER_ROLE");
      bytes32 public constant TRANSFER_ROLE = keccak256("TRANSFER_ROLE");
  
      mapping(uint256 => bool) public processed;
      mapping(uint256 => uint256) public chainToFrom;
  
      mapping(uint256 => bool) private _tokenWhitelist;
  
      mapping(address => bool) public blacklist;
  
      bool public forwardNative = true;
  
      constructor(address admin, address _liquidityPool) {
          liquidityPool = _liquidityPool;
  
          _grantRole(DEFAULT_ADMIN_ROLE, admin);
          _grantRole(ADMIN_ROLE, admin);
          _grantRole(PAUSER_ROLE, admin);
          _grantRole(UNPAUSER_ROLE, admin);
          _grantRole(UNLOCKER_ROLE, admin);
          _grantRole(TRANSFER_ROLE, admin);
  
      }
  
      receive() external payable whenNotPaused {}
  
  
      event Lock(uint256 indexed fromToken, uint256 indexed toToken, address indexed fromAccount, address toAccount, uint256 fromAmount, string metadata);
      event Unlock(uint256 indexed cce, uint256 indexed toToken, address indexed toAccount, uint256 toAmount);
  
      event AdminTransfer(address indexed token, address recipient, uint256 amount);
  
      event SetLiquidityPool(address pool);
      event SetProcessed(uint256 cce, bool value);
  
      event SetTokenWhitelist(uint256 fromToken, bool value);
  
      event SetBlacklist(address user, bool value);
  
      event SetForwardNative(bool value);
  
      function tokenWhitelist(address token) external view returns(bool) {
          return _tokenWhitelist[toCCA(block.chainid, token)];
      }
  
      function tokenWhitelist(uint256 cca) external view returns(bool) {
          return _tokenWhitelist[cca];
      }
  
      function pause() public onlyRole(PAUSER_ROLE) {
          _pause();
      }
  
      function unpause() public onlyRole(UNPAUSER_ROLE) {
          _unpause();
      }
  
      function toCCA(uint256 chainId, address localAddress) public pure returns(uint256 cca) {
          return (chainId << 160) + uint256(uint160(localAddress));
      }
  
      function fromCCA(uint256 cca) public pure returns(uint256 chainId, address localAddress) {
          chainId = cca >> 160;
          localAddress = address(uint160(cca));
      }
  
      function toCCE(uint64 chainId, uint64 blockNumber, uint64 transactionIndex, uint64 logIndex) public pure returns(uint256 cce) {
          uint256 bits = 64;
  
          return (uint256(chainId) << (3*bits))
              + (uint256(blockNumber) << (2*bits))
              + (uint256(transactionIndex) << (bits))
              + uint256(logIndex);
      }
  
      function fromCCE(uint256 cce) public pure returns(uint64 chainId, uint64 blockNumber, uint64 transactionIndex, uint64 logIndex) {
          uint256 bits = 64;
  
          chainId = uint64(cce >> (3*bits));
          blockNumber = uint64(cce >> (2*bits));
          transactionIndex = uint64(cce >> bits);
          logIndex = uint64(cce);
      }
  
      // const { fromTokenAddress, toTokenAddress, amount } = event.returnValues;
      function _lock(uint256 fromToken, uint256 toToken, address fromAccount, address toAccount, uint256 fromAmount, string calldata metadata) private {
  
          ( /* uint256 chainId */, address fromTokenAddress) = fromCCA(fromToken);
  
          require(_tokenWhitelist[fromToken], "Token must be whitelisted");
  
          require(!blacklist[fromAccount] && !blacklist[toAccount] && !blacklist[msg.sender], "Accounts cannot be blacklisted");
  
          require(fromAccount != address(0) && toAccount != address(0), "Accounts must not be the null address");
  
          require(fromAmount > 0, "Amount must be greater than 0");
  
          require(fromTokenAddress != address(0) || fromAmount == msg.value, "msg.value must equal amount");
  
  
          if(fromTokenAddress == address(0)) {
              if(forwardNative) payable(liquidityPool).transfer(msg.value);
          } else {
              IERC20(fromTokenAddress).safeTransferFrom(fromAccount, liquidityPool, fromAmount);
          }
  
          emit Lock(fromToken, toToken, fromAccount, toAccount, fromAmount, metadata);
      }
  
      function lock(uint256 fromToken, uint256 toToken, uint256 amount, string calldata metadata) external payable whenNotPaused {
          _lock(fromToken, toToken, msg.sender, msg.sender, amount, metadata);
      }
  
      function lock(uint256 fromToken, uint256 toToken, address fromAccount, address toAccount, uint256 amount, string calldata metadata) external payable whenNotPaused {
          _lock(fromToken, toToken, fromAccount, toAccount, amount, metadata);
      }
  
      function unlock(uint256 cce, uint256 token, address account, uint256 amount) public payable onlyRole(UNLOCKER_ROLE) returns(bool) {
          bool isProcessed = processed[cce];
          if(isProcessed) return false;
  
          ( /* uint256 chainId */, address tokenAddress) = fromCCA(token);
  
          require(!blacklist[account], "Accounts cannot be blacklisted");
          require(account != address(0), "Accounts must not be the null address");
  
          require(amount > 0, "Amount must be greater than 0");
  
          if(tokenAddress == address(0)) {
              payable(account).transfer(amount);
          } else {
              // reverts on failure
              IERC20(tokenAddress).safeTransferFrom(liquidityPool, account, amount);
          }
  
          processed[cce] = true;
  
          uint256 fromChainId = cce >> 3 * 64;
          uint256 currentFrom = chainToFrom[fromChainId];
  
          if(currentFrom < cce) chainToFrom[fromChainId] = cce;
  
          emit Unlock(cce, token, account, amount);
          return true;
      }
  
      // these are for extracting funds from the contract for when users mistakenly send funds directly to it
      function transferNative(address payable recipient, uint256 amount) external onlyRole(TRANSFER_ROLE) {
          recipient.transfer(amount);
          emit AdminTransfer(address(0), recipient, amount);
      }
  
      function transferERC20(address token, address recipient, uint256 amount) external onlyRole(TRANSFER_ROLE) {
          IERC20(token).safeTransfer(recipient, amount);
          emit AdminTransfer(token, recipient, amount);
      }
  
  
      function setLiquidityPool(address pool) external onlyRole(ADMIN_ROLE) {
          liquidityPool = pool;
          emit SetLiquidityPool(pool);
      }
  
      function setProcessed(uint256 cce, bool value) external onlyRole(ADMIN_ROLE) {
          processed[cce] = value;
          emit SetProcessed(cce, value);
      }
  
      function setTokenWhitelist(uint256 token, bool value) external onlyRole(ADMIN_ROLE) {
          _tokenWhitelist[token] = value;
          emit SetTokenWhitelist(token, value);
      }
  
      function setBlacklist(address user, bool value) external onlyRole(ADMIN_ROLE) {
          blacklist[user] = value;
          emit SetBlacklist(user, value);
      }
  
      function setForwardNative(bool value) external onlyRole(ADMIN_ROLE) {
          forwardNative = value;
          emit SetForwardNative(value);
      }
  }